Lecture 4 – Types

- Every language has types – only difference is when to check them.
  - Check types before runtime: static language
  - Check types during runtime: dynamic language
  - Static language has a compiler: part of its job is to check (and infer) types.
- A type is a set of values – the (declared/inferred) type of an expression specifies its range of values.

Primitive and constructed types

- Primitive types are predefined by the language:

```
Int, Char, Bool, Integer, Float, Double
```

- Constructed types are defined by composing type constructors and primitive types:

```
[Int]    (Int, Bool)    Int -> Bool
```

- Type constructors include

```
[]     (,)      ->
```

- String is a constructed type: [Char]
- Find Haskell type with `:t`.

Type annotation

Every declaration can be annotated with a type.

```
1  x :: Int
2  x = 1
3
4  x' :: Integer
5  x' = 1
6
7  fact :: Int -> Int
8  fact 0 = 1
9  fact n = n * fact(n-1)
10
11 fact' :: Integer -> Integer
12 fact' 0 = 1
13 fact' n = n * fact(n-1)
```

Type variable and polymorphism

A function type that contains type variables is a polytype.
The type variables in a polytype can be instantiated with other types.

```
1  f x = fst x
2
3  :t f     -- f :: (a, b) -> a
4
5  f (1, True) -- evaluates to 1
6
7  f (True, 1) -- evaluates to True
```

Type variable and polymorphism

Functions with polytypes allow polymorphism.

```
1  map :: (a -> b) -> [a] -> [b]
2
3  foldl :: (b -> a -> b) -> b -> [a] -> b
4
5  foldr :: (a -> b -> b) -> b -> [a] -> b
6
7  map (\x -> x + 1) :: [Int] -> [Int]
8
9  foldl (\c e -> c && e > 0) :: Bool -> [Int] -> Bool
10
11 foldr (:) :: [a] -> [a] -> [a]
```

Algebraic Data Type (ADT)

We can define new type constructors and data constructors using algebraic data types.

```
1 data Coin = Head | Tail deriving (Show)
2
3 t: Head -- Coin
4
5 coinFlip Head = Tail
6 coinFlip Tail = Head
7
8 t: coinFlip -- Coin -> Coin
```

`Coin` is a type constructor (or a type if no parameters)
`Head` and `Tail` are data constructors (or data if no parameters).
`deriving (Show)` auto-derives functions for `Showing` the data.

ADT for Enumeration

```
1  data WeekDay = Tuesday | OpenDay | Thursday
2               | MeetingDay | BlurDays deriving (Show)
3
4  nextDay Tuesday = OpenDay
5  nextDay OpenDay = Thursday
6  nextDay Thursday = MeetingDay
7  nextDay MeetingDay = BlurDays
8  nextDay BlurDays = Tuesday
9
10 :t nextDay -- WeekDay -> WeekDay
```

ADT for Enumeration

```
1 data WeekDay = Tuesday | OpenDay | Thursday
2              | MeetingDay | BlurDays deriving (Show, Eq)
3
4 canWeMeet day
5    | day == OpenDay = True
6    | otherwise = False
```

In order to use == on WeekDay/DayTime typed data, we need to derive operators of Eq class.

ADT for Enumeration

```
1  data WeekDay = Tuesday | OpenDay | Thursday
2               | MeetingDay | BlurDays deriving (Show, Eq)
3
4  data DayTime = Morning | Afternoon | Night deriving (Eq)
5
6  canWeMeet day time
7      | day == OpenDay = True
8      | day == Tuesday && time == Morning = True
9      | day == Thursday && time == Morning = True
10     | otherwise = False
```

In order to use == on WeekDay typed data, we need to derive operators of Eq class.

ADT for Wrapped Data

Data constructor can wrap parameters.

```
1  data MeetTime = Meet WeekDay DayTime deriving (Show)
2
3  :t Meet Tuesday Morning     -- MeetTime
4
5  schedule day time
6     | canWeMeet day time = Meet day time
7     | otherwise = schedule (nextDay day) time
8
9  schedule Tuesday Morning     -- Meet Tuesday Morning
10 schedule Tuesday Afternoon   -- Meet OpenDay Afternoon
11 schedule Thursday Afternoon -- Meet OpenDay Afternoon
```

ADT for Wrapped Data

A data type can be used to create a distinguished value.
WeekNum is more than just a number.

```
1  data WeekNum = Week Int deriving (Show, Eq)
2  nextWeek (Week n) = Week (n+1)
3
4  data MeetTime = Meet WeekNum WeekDay DayTime deriving (Show, Eq)
5
6  schedule (Meet week day time)
7     | canWeMeet day time = Meet week day time
8     | otherwise = schedule meet'
9
10    where meet' = Meet week' day' time
11          day' = nextDay day
12          week' = if (day' /= day && day' == Tuesday)
13                  then nextWeek week
14                  else week
15
16 m1 = Meet (Week 1) Thursday Afternoon
17 schedule m1      -- Meet (Week 2) OpenDay Afternoon
```

Mix builtin types with ADT

We can represent a schedule as a list of MeetTime values.

```
1  schedule meetings meet
2     | canMeet && noConflict = ((meet:meetings), meet)
3     | otherwise = schedule meetings meet'
4
5     where (Meet week day time) = meet
6
7           canMeet = canWeMeet day time
8           noConflict = not (meet `elem` meetings)
9
10          day' = nextDay day
11          week' = if (day' /= day && day' == Tuesday)
12                    then nextWeek week
13                    else week
14          meet' = Meet week' day' time
```

Mix builtin types with ADT

We can represent a schedule as a list of MeetTime values.

```
showSchedule s = foldl (++) "" $ map (\m -> show m ++ ";\n") s

m1 = Meet (Week 1) Tuesday Afternoon
m2 = Meet (Week 1) OpenDay Morning
m3 = Meet (Week 1) Thursday Morning
(s1, m1') = schedule [] m1
(s2, m2') = schedule s1 m2
(s3, m3') = schedule s2 m3
putStrLn $ "current schedule is: \n" ++ showSchedule s3

-- current schedule is:
-- Meet (Week 1) Thursday Morning;
-- Meet (Week 1) OpenDay Morning;
-- Meet (Week 1) OpenDay Afternoon;
```

Order meetings

Define insert function to ensure that meetings are ordered.

```
1 insertMeet meet [] = [meet]
2 insertMeet meet (a:b)
3     | meet <= a = meet:a:b
4     | otherwise = a : insertMeet meet b
```

However, to compare meet using $\leq$, we have to derive `Ord` instance for MeetTime.

Order meetings

All types involved in MeetTime has to be instances of `Ord`.

```
1  data WeekDay = Tuesday | OpenDay | Thursday
2               | MeetingDay | BlurDays deriving (Show, Eq, Ord)
3
4  data DayTime = Morning | Afternoon | Night
5                                     deriving (Show, Eq, Ord)
6
7  data WeekNum = Week Int deriving (Show, Eq, Ord)
8
9  data MeetTime = Meet WeekNum WeekDay DayTime
10                                    deriving (Show, Eq, Ord)
```

Order meetings

Revise schedule function to insert meet to the current schedule.

```
1  schedule meetings meet
2     | canMeet && noConflict = (insertMeeting meet meetings, meet)
3     | otherwise = schedule meetings meet'
4
5     where (Meet week day time) = meet
6
7           canMeet = canWeMeet day time
8           noConflict = not (meet `elem` meetings)
9
10          day' = nextDay day
11          week' = if (day' /= day && day' == Tuesday)
12                    then nextWeek week
13                    else week
14          meet' = Meet week' day' time
```

Order meetings

Now the scheduled meetings are ordered.

```
showSchedule s = foldl (++) "" $ map (\m -> show m ++ ";\n") s

m1 = Meet (Week 1) Tuesday Afternoon
m2 = Meet (Week 1) OpenDay Morning
m3 = Meet (Week 1) Thursday Morning
(s1, m1') = schedule [] m1
(s2, m2') = schedule s1 m2
(s3, m3') = schedule s2 m3
putStrLn $ "current schedule is: \n" ++ showSchedule s3

-- current schedule is:
-- Meet (Week 1) OpenDay Morning;
-- Meet (Week 1) OpenDay Afternoon;
-- Meet (Week 1) Thursday Morning;
```

Recursive ADT

ADT (Algebraic Data Type) can be defined recursively.
An event type can be defined as a labelled tree with two types of nodes:

- Tag node that has tag string and a list of events
- One node that has a meeting time.

```
1 data Event = Tag String [Event]
2              | One MeetTime          deriving (Show)
```

We can represent a person's calendar using the type `[Event]`

Type alias

Type alias can improve readability and clarify intention.
The calendar type below is an alias of the type [Event]

```
1 type Calendar = [Event]
2
3 data Event = Tag String Calendar
4            | One MeetTime              deriving (Show)
```

Work with recursive ADT

The getMeetings function extracts a list of meetings from a calendar.

```
1 type Calendar = [Event]
2
3 data Event = Tag String Calendar
4            | One MeetTime              deriving (Show)
5
6 -- convert a calendar to a list of meetings
7 getMeetings calendar = foldr (++) [] $ map get calendar
8     where get (One meet) = [meet]
9           get (Tag _ calendar') = getMeetings calendar'
```

Work with recursive ADT

The makeEvent function creates an event from a list of tags and a meeting time.

```
type Calendar = [Event]

data Event = Tag String Calendar
           | One MeetTime                deriving (Show)

-- make an event from a list of tags and a meeting time
makeEvent [] meet = One meet
makeEvent (tag:tags) meet = Tag tag [makeEvent tags meet]

m1 = Meet (Week 1) Tuesday Afternoon
makeEvent ["Courses", "790", "Bob"] m1 -- evaluates to
-- Tag "Courses"
--   [Tag "790"
--      [Tag "Bob"
--         [One (Meet (Week 1) Tuesday Afternoon)]]]
```

Work with recursive ADT

The insert function inserts an event (denoted by a list of tags and meeting time) to a calendar.

```
1  type Calendar = [Event]
2
3  data Event = Tag String Calendar
4             | One MeetTime             deriving (Show)
5
6  -- insert an event to a calendar
7  insert [] tags meet = [makeEvent tags meet]
8
9  insert ((One _) : events) tags meet = insert events tags meet
10
11 insert ((Tag tag calendar): events) (tag':tags') meet
12
13  | tag == tag' = (Tag tag $ insert calendar tags' meet) : events
14
15  | otherwise = event : (insert events tags meet)
16
17    where event = (Tag tag calendar)
18          tags = (tag' : tags')
```

Work with recursive ADT

The addEvent function adds an event to a calendar (also ensures no time conflict).

```
1  type Calendar = [Event]
2
3  data Event = Tag String Calendar
4             | One MeetTime             deriving (Show)
5
6  -- calendar is a list of events
7  -- tags is a list, where each tag is a string
8  -- meet is a meeting time
9  -- calendar' is the new list of events
10 -- meet' is the scheduled meeting time
11 addEvent calendar tags meet = (calendar', meet')
12      where meetings = getMeetings calendar
13            (_, meet') = schedule meetings meet
14            calendar' = insert calendar tags meet'
```

Work with recursive ADT

The showCalendar function prints each event in a calendar in a separate line and use tabs to align events of the same depth in the calendar.

```
1  type Calendar = [Event]
2
3  data Event = Tag String Calendar
4             | One MeetTime              deriving (Show)
5
6  -- format calendar
7  showCalendar n calendar =
8      foldl (++) "" $ map (\e -> show' n e ++ "\n") calendar
9
10   where show' n (One m) = tabs n ++ show m
11
12         show' n (Tag tag lst) =
13            tabs n ++ tag ++ "\n" ++ (showCalendar (n+1) lst)
14
15         tabs n = take n $ repeat '\t'
```

Work with recursive ADT

```
1  m1 = Meet (Week 1) Tuesday Afternoon
2  m2 = Meet (Week 1) OpenDay Morning
3  m3 = Meet (Week 1) Thursday Morning
4  (c1, _) = addEvent [] ["Research", "Adam"] m1
5  (c2, _) = addEvent c1 ["Research", "Alan"] m2
6  (c3, _) = addEvent c2 ["Courses", "790", "Bob"] m3
7  (c4, _) = addEvent c3 ["Courses", "431", "Carol"] m3
8  putStrLn $ showCalendar 0 c4
9
10 -- Research
11 --       Adam
12 --                Meet (Week 1) OpenDay Afternoon
13 --
14 --       Alan
15 --                Meet (Week 1) OpenDay Morning
16 --
17 -- Courses
18 --       790
19 --                Bob
20 --                        Meet (Week 1) Thursday Morning
21 --
22 --       431
23 --                Carol
24 --                        Meet (Week 2) Tuesday Morning
```